Pensieve header: The full sl_2 invariant using the Drinfel'd double. Continues 2018-05/ybax.nb, Talks/StonyBrook-1805/ybax.nb, Projects/SL2Portfolio/Logoi.nb.

Profiling

In[ ]:=
```
Once[
  SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\SL2Invariant"];
  << KnotTheory`;
  << "../Profile/Profile.m";
 ];
BeginProfile[];
Once@PopupWindow[Button["Show Profile Monitor"],
  Dynamic[PrintProfile[], UpdateInterval → 3, TrackedSymbols → {}]]
```

ParentDirectory: Argument File should be a positive machine-size integer, a nonempty string, or a File specification.

ParentDirectory: Argument File should be a positive machine-size integer, a nonempty string, or a File specification.

ToFileName: String or list of strings expected at position 1 in ToFileName[{File, WikiLink, mathematica}].

ToFileName: String or list of strings expected at position 1 in ToFileName[{File, QuantumGroups}].

```
Loading KnotTheory` version of January 20, 2015, 10:42:19.1122.
Read more at http://katlas.org/wiki/KnotTheory.

This is Profile.m of http://www.drorbn.net/AcademicPensieve/Projects/Profile/.

This version: June 2018. Original version: July 1994.
```

Out[ ]= Show Profile Monitor

External Utilities

In[ ]:=
```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Program

Program

Internal Utilities

Program

Canonical Form:

Program

In[∘]:=
```
CCF[ℰ_] := PP_CCF@ExpandDenominator@ExpandNumerator@PP_Together@Together[PP_Exp[
        Expand[ℰ] //. e^x_ e^y_ :> e^(x+y) /. e^x_ :> e^CCF[x]]];
CF[ℰ_List] := CF /@ ℰ;
CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := PP_CF@Module[
    {vs = Cases[ℰ, (y | b | t | a | x | η | β | τ | α | ξ)_, ∞] ⋃ {y, b, t, a, x, η, β, τ, α, ξ}},
    Total[CoefficientRules[Expand[ℰ], vs] /. (ps_ → c_) :> CCF[c] (Times @@ vs^ps)]
   ];
CF[ℰ_𝔼] := CF /@ ℰ; CF[𝔼_sp___[ℰs___]] := CF /@ 𝔼_sp[ℰs];
```

Program

The Kronecker δ:

Program

In[∘]:=
```
Kδ /: Kδ_{i_,j_} := If[i === j, 1, 0];
```

Program

Equality, multiplication, and degree-adjustment of perturbed Gaussians; $\mathbb{E}[L, Q, P]$ stands for $e^{L+Q} P$:

Program

In[∘]:=
```
𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
   CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] := 𝔼[L1 + L2, Q1 + Q2, P1 * P2];
𝔼[L_, Q_, P_]_$k_ := 𝔼[L, Q, Series[Normal@P, {ϵ, 0, $k}]];
```

Program

In[∘]:=
```
𝔼3@𝔼[ω_, L_, Q_, Ps_] := CF /@ 𝔼[L, ω^-1 Q, ω^-1 (ω^-4 ϵ)^(-1+Range@Length@Ps).Ps]_$k;
𝔼4@𝔼[L_, Q_, P_] := Module[
    {ω = Normal[P]^-1 /. ϵ → 0, Ps = CoefficientList[P, ϵ]},
    CF /@ 𝔼[ω, L, ω Q, ω^(-3+4 Range@Length@Ps) Ps]];
𝔼3@𝔼_sp___[as___] := 𝔼3@𝔼[as] /. 𝔼 → 𝔼_sp;
𝔼4@𝔼_sp___[as___] := 𝔼4@𝔼[as] /. 𝔼 → 𝔼_sp;
```

Program

Zip and Bind

Program

Variables and their duals:

Program

In[∘]:=
```
{t*, b*, y*, a*, x*, z*} = {τ, β, η, α, ξ, ζ};
{τ*, β*, η*, α*, ξ*, ζ*} = {t, b, y, a, x, z}; (u_i_)* := (u*)_i;
```

Program

Finite Zips:

Program

In[∘]:=
```
collect[sd_SeriesData, ζ_] := MapAt[collect[#, ζ] &, sd, 3];
collect[ℰ_, ζ_] := PP_Collect@Collect[ℰ, ζ];
Zip_{}[P_] := P;
Zip_ζs_[Ps_List] := Zip_ζs /@ Ps;
Zip_{ζ_,ζs___}[P_] := PP_Zip[
  (collect[P // Zip_{ζs}, ζ] /. f_. ζ^d_. ⧴ ∂_{ζ*,d} f) /. ζ* → 0]
```

Program

QZip implements the “Q-level zips” on $\mathbb{E}(L, Q, P) = e^{L+Q} P(\epsilon)$ and/or on $\mathbb{E}(\omega, L, Q, P) = \omega^{-1} e^{L+\omega^{-1} Q} P(\omega^{-4} \epsilon)$. Such zips regard the L variables as scalars.

$$\left\langle P(z_i,\zeta^j)e^{c+\eta^i z_i+y_j\zeta^j+q^i_j z_i\zeta^j}\right\rangle = |\tilde{q}| \left\langle P(z_i,\zeta^j)e^{c+\eta^i z_i}\Big|_{z_i\to\tilde{q}^k_i(z_k+y_k)}\right\rangle$$
$$= |\tilde{q}|e^{c+\eta^i\tilde{q}^k_i y_k}\left\langle P\left(\tilde{q}^k_i(z_k+y_k),\zeta^j+\eta^i\tilde{q}^j_i\right)\right\rangle.$$

Program

In[∘]:=
```
$QZipFail = False;
QZip_ζs_List@𝔼[L_, Q_, P_] := PP_QZip@Module[{ζ, z, zs, c, ys, ηs, qt, zrule, ζrule, out},
    zs = Table[ζ*, {ζ, ζs}];
    c = CF[Q /. Alternatives @@ (ζs ⋃ zs) → 0];
    ys = CF@Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
    ηs = CF@Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
    qt = CF@Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ}Q, {ζ, ζs}, {z, zs}];
    zrule = Thread[zs → CF[qt.(zs + ys)]];
    ζrule = Thread[ζs → ζs + ηs.qt];
    out = CF /@ 𝔼[L, c + ηs.qt.ys, Det[qt] Zip_ζs[P /. (zrule ⋃ ζrule)]];
    If[¬ ($QZipFail ∨ TrueQ[out ≡ 𝔼3@QZip_ζs@𝔼4@𝔼[L, Q, P]]),
     $QZipFail = True; Print["QZip4 fail at {L,Q,P}=", {L, Q, P}]
    ];
    out
   ];
```

Program

In[∘]:=
```
$QZipFail = False;
QZip_ζs_List@𝔼[ω_, L_, Q_, Ps_] := PP_QZip4@Module[{ζ, z, zs, c, ys, ηs, qt, zrule, ζrule},
    zs = Table[ζ*, {ζ, ζs}];
    c = CF[Q /. Alternatives @@ (ζs ⋃ zs) → 0];
    ys = CF@Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
    ηs = CF@Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
    qt = CF@Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ}Q, {ζ, ζs}, {z, zs}];
    zrule = Thread[zs → CF[qt.(zs + ys)]];
    ζrule = Thread[ζs → ζs + ηs.qt];
    CF /@ 𝔼[ω Det[qt / ω], L, c + ηs.qt.ys, Zip_ζs[Ps /. (zrule ⋃ ζrule)]] ];
```

Program

Upper to lower and lower to Upper:

Program

In[ ]:=
```
U2l = {B_i_^p_. → e^(-p ħ γ b_i), B^p_. → e^(-p ħ γ b), T_i_^p_. → e^(p ħ t_i), T^p_. → e^(p ħ t), 𝒜_i_^p_. → e^(p γ α_i), 𝒜^p_. → e^(p γ α)};
l2U = {e^(c_. b_i_+d_.) :> B_i^(-c/(ħ γ)) e^d, e^(c_. b+d_.) :> B^(-c/(ħ γ)) e^d,
    e^(c_. t_i_+d_.) :> T_i^(c/ħ) e^d, e^(c_. t+d_.) :> T^(c/ħ) e^d,
    e^(c_. α_i_+d_.) :> 𝒜_i^(c/γ) e^d, e^(c_. α+d_.) :> 𝒜^(c/γ) e^d,
    e^ℰ_ :> e^Expand@ℰ};
```

Program

LZip implements the "L-level zips" on $\mathbb{E}(L, Q, P) = P e^{L+Q}$. Such zips regard all of $P e^Q$ as a single"P". Here the z's are b and α and the ζ's are β and a.

Program

In[ ]:=
```
LZip_ζs_List@𝔼[L_, Q_, P_] :=
  PP_LZip@Module[{ζ, z, zs, c, ys, ηs, lt, zrule, Zrule, ζrule, Q1, EEQ, EQ},
    zs = Table[ζ*, {ζ, ζs}];
    c = L /. Alternatives @@ (ζs ⋃ zs) → 0;
    ys = Table[∂_ζ (L /. Alternatives @@ zs → 0), {ζ, ζs}];
    ηs = Table[∂_z (L /. Alternatives @@ ζs → 0), {z, zs}];
    lt = Inverse@Table[Kδ_(z,ζ*) - ∂_(z,ζ) L, {ζ, ζs}, {z, zs}];
    zrule = Thread[zs → lt.(zs + ys)];
    Zrule = zrule /. r_Rule :>
       ((U = r⟦1⟧ /. {b → B, t → T, α → 𝒜}) → (U /. U2l /. r //. l2U)); (* not used *)
    ζrule = Thread[ζs → ζs + ηs.lt];
    Q1 = Q /. U2l /. (zrule ⋃ ζrule);
    EEQ[ps___] := EEQ[ps] = PP_"EEQ"@(CF[e^(-Q1) D[e^Q1, Sequence @@ Thread[{zs, {ps}}]]] /.
          Alternatives @@ zs → 0 //. l2U);
    CF /@ ((*CF/@*)𝔼[
        c + ηs.lt.ys, Q1 /. Alternatives @@ zs → 0,
        Det[lt] (Zip_ζs[(EQ @@ zs) (P /. U2l /. (zrule ⋃ ζrule))] /.
           Derivative[ps___][EQ][___] :> EEQ[ps] /. _EQ → 1)
      ] //. l2U)
   ];
```

Program

In[ ]:=
```
B_{}[L_, R_] := L R;
B_{is__}[L_𝔼, R_𝔼] := PP_B@Module[{n},
    Times[
      L /. Table[(v : b | B | t | T | a | x | y)_i → v_n@i, {i, {is}}],
      R /. Table[(v : β | τ | α | 𝒜 | ξ | η)_i → v_n@i, {i, {is}}]
     ] // LZip_Join@@Table[{β_n@i,τ_n@i,a_n@i},{i,{is}}] // QZip_Join@@Table[{ξ_n@i,y_n@i},{i,{is}}] ];
B_is___[L_, R_] := B_{is}[L, R];
```

Program

𝔼 morphisms with domain and range.

Program

In[∘]:=
```
B_is_List[𝔼_{d1_→r1_}[L1_, Q1_, P1_], 𝔼_{d2_→r2_}[L2_, Q2_, P2_]] :=
  𝔼_{(d1⋃Complement[d2,is])→(r2⋃Complement[r1,is])} @@ B_is[𝔼[L1, Q1, P1], 𝔼[L2, Q2, P2]];
𝔼_{d1_→r1_}[L1_, Q1_, P1_] // 𝔼_{d2_→r2_}[L2_, Q2_, P2_] :=
  B_{r1⋂d2}[𝔼_{d1→r1}[L1, Q1, P1], 𝔼_{d2→r2}[L2, Q2, P2]];
𝔼_{d1_→r1_}[L1_, Q1_, P1_] ≡ 𝔼_{d2_→r2_}[L2_, Q2_, P2_] ^:=
  (d1 == d2) ∧ (r1 == r2) ∧ (𝔼[L1, Q1, P1] ≡ 𝔼[L2, Q2, P2]);
𝔼_{d1_→r1_}[L1_, Q1_, P1_] 𝔼_{d2_→r2_}[L2_, Q2_, P2_] ^:=
  𝔼_{(d1⋃d2)→(r1⋃r2)} @@ (𝔼[L1, Q1, P1] 𝔼[L2, Q2, P2]);
𝔼_{d_→r_}[L_, Q_, P_]_{$k_} := 𝔼_{d→r} @@ 𝔼[L, Q, P]_{$k};
𝔼_[ℰ___][i_] := {ℰ}[[i]];
```

Program

“Define” code

Program

Define[lhs = rhs,   ...] defines the lhs to be rhs, except that rhs is computed only once for each value of $k. Fancy Mathematica not for the faint of heart. Most readers should ignore.

Program

In[∘]:=
```
SetAttributes[Define, HoldAll];
Define[def_, defs__] := (Define[def]; Define[defs];);
Define[op_{is__} = ℰ_] := Module[{SD, ii, jj, kk, isp, nis, nisp, sis}, Block[{i, j, k},
    ReleaseHold[Hold[
        SD[op_{nisp,$k_Integer}, PP_Boot@Block[{i, j, k}, op_{isp,$k} = ℰ; op_{nis,$k}]];
        SD[op_isp, op_{{is},$k}]; SD[op_{sis__}, op_{{sis}}];
      ] /. {SD → SetDelayed,
        isp → {is} /. {i → i_, j → j_, k → k_},
        nis → {is} /. {i → ii, j → jj, k → kk},
        nisp → {is} /. {i → ii_, j → jj_, k → kk_}
      }] ]]
```

Program

Booting Up

Program

In[∘]:=
```
$k = 2; (*ℏ=γ=1;*)
```

Program

In[∘]:=
```
Define[am_{i,j→k} = 𝔼_{{i,j}→{k}}[(α_i + α_j) a_k, (e^{-γ α_j} ξ_i + ξ_j) x_k, 1]_{$k},
 bm_{i,j→k} = 𝔼_{{i,j}→{k}}[(β_i + β_j) b_k, (η_i + η_j) y_k, e^{(e^{-ϵ β_i}-1) η_j y_k}]_{$k}]
```

Program

In[∘]:=
```
Define[R_{i,j} = CF@𝔼_{{}→{i,j}}[ℏ a_j b_i, ℏ x_j y_i, ⅇ^(Sum_{k=2}^{$k+1} ((1 - ⅇ^{γ ϵ ℏ})^k (ℏ y_i x_j)^k)/(k (1 - ⅇ^{k γ ϵ ℏ})))]_{$k},
  R̄_{i,j} = CF@𝔼_{{}→{i,j}}[-ℏ a_j b_i, -ℏ x_j y_i / B_i, 1 + If[$k == 0, 0, (R̄_{{i,j},$k-1})_{$k}[3] -
      (((R̄_{{i,j},0})_{$k} R_{1,2} (R̄_{{3,4},$k-1})_{$k}) // (bm_{i,1→i} am_{j,2→j}) // (bm_{i,3→i} am_{j,4→j}))[3]]],
  P_{i,j} = 𝔼_{{i,j}→{}}[β_i α_j / ℏ, η_i ξ_j / ℏ, 1 + If[$k == 0, 0, (P_{{i,j},$k-1})_{$k}[3] -
      (R_{1,2} // ((P_{{1,j},0})_{$k} (P_{{i,2},$k-1})_{$k}))[3]]]]
```

Program

In[∘]:=
```
Define[aS_j = R̄_{i,j} ~ B_i ~ P_{i,j},
  aS̄_i = 𝔼_{{i}→{i}}[-a_i α_i, -x_i 𝒜_i ξ_i, 1 + If[$k == 0, 0, (aS̄_{{i},$k-1})_{$k}[3] -
      ((aS̄_{{i},0})_{$k} ~ B_i ~ aS_i ~ B_i ~ (aS̄_{{i},$k-1})_{$k})[3]]]]
```

Program

In[∘]:=
```
Define[bS_i = R_{i,1} ~ B_1 ~ aS_1 ~ B_1 ~ P_{i,1},
  bS̄_i = R_{i,1} ~ B_1 ~ aS̄_1 ~ B_1 ~ P_{i,1},
  aΔ_{i→j,k} = (R_{1,j} R_{2,k}) // bm_{1,2→3} // P_{3,i},
  bΔ_{i→j,k} = (R_{j,1} R_{k,2}) // am_{1,2→3} // P_{i,3}]
```

Program

In[∘]:=
```
Define[dm_{i,j→k} = (𝔼_{{i,j}→{i,j}}[β_i b_i + α_j a_j, η_i y_i + ξ_j x_j, 1]
      (aΔ_{i→1,2} // aΔ_{2→2,3} // aS̄_3) (bΔ_{j→-1,-2} // bΔ_{-2→-2,-3})) // (P_{-1,3} P_{-3,1} am_{2,j→k} bm_{i,-2→k}),
  dS_i = 𝔼_{{i}→{1,2}}[β_i b_1 + α_i a_2, η_i y_1 + ξ_i x_2, 1] // (bS̄_1 aS_2) // dm_{2,1→i},
  dΔ_{i→j,k} = (bΔ_{i→3,1} aΔ_{i→2,4}) // (dm_{3,4→k} dm_{1,2→j})]
```

Program

In[∘]:=
```
Define[C_i = 𝔼_{{}→{i}}[0, 0, B_i^{1/2} ⅇ^{-ℏ ϵ a_i/2}]_{$k},
  C̄_i = 𝔼_{{}→{i}}[0, 0, B_i^{-1/2} ⅇ^{ℏ ϵ a_i/2}]_{$k},
  Kink_i = (R_{1,3} C̄_2) // dm_{1,2→1} // dm_{1,3→i},
  K̄ink_i = (R̄_{1,3} C_2) // dm_{1,2→1} // dm_{1,3→i}]
```

Program

Note. $t == \epsilon a - \gamma b$ and $b == -t/\gamma + \epsilon a/\gamma$.

Program

In[∘]:=
```
Define[b2t_i = 𝔼_{{i}→{i}}[α_i a_i - β_i t_i / γ, ξ_i x_i + η_i y_i, ⅇ^{ϵ β_i a_i/γ}]_{$k},
  t2b_i = 𝔼_{{i}→{i}}[α_i a_i - τ_i γ b_i, ξ_i x_i + η_i y_i, ⅇ^{ϵ τ_i a_i}]_{$k}]
```

Testing

In[∘]:=
```
Block[{$k = 1}, {
   am → am_{i,j→k}, bm → bm_{i,j→k}, dm → dm_{i,j→k}, R → R_{i,j}, R̄ → R̄_{i,j}, P → P_{i,j},
   aS → aS_i, aS̄ → aS̄_i, bS → bS_i, bS̄ → bS̄_i, dS → dS_i, aΔ → aΔ_{i→j,k}, bΔ → bΔ_{i→j,k},
   dΔ → dΔ_{i→j,k}, C → C_i, C̄ → C̄_i, Kink → Kink_i, Kink̄ → Kink̄_i, b2t → b2t_i, t2b → t2b_i
  }] //
 Column
```

QZip4 fail at $\{L,Q,P\}=\Big\{\hbar\, a_3\, b_1,$

$\hbar\, x_3\, y_{n\$13472[1]} + y_1\, \eta_{n\$13472[1]} + y_1\, \eta_{n\$13472[2]} + x_1\, \xi_{n\$13472[1]} + \frac{(1-B_1)\, \eta_{n\$13472[2]}\, \xi_{n\$13472[1]}}{\hbar} + x_1\, \xi_{n\$13472[2]},$

$$\frac{1}{\sqrt{B_1}} + \left(\frac{\hbar\, a_1}{2\sqrt{B_1}} - \frac{\gamma\, \hbar^3\, x_3^2\, y_{n\$13472[1]}^2}{4\sqrt{B_1}} - \frac{\hbar\, a_3\, y_1\, \eta_{n\$13472[2]}}{\sqrt{B_1}} - \frac{\gamma\, \hbar\, x_1\, \xi_{n\$13472[1]}}{\sqrt{B_1}} + a_1\sqrt{B_1}\, \eta_{n\$13472[2]}\, \xi_{n\$13472[1]} + \frac{\gamma\, \hbar\, x_1\, y_1\, \eta_{n\$13472[2]}\, \xi_{n\$13472[1]}}{\sqrt{B_1}} + \frac{(\gamma - 3\gamma B_1)\, y_1\, \eta_{n\$13472[2]}^2\, \xi_{n\$13472[1]}}{2\sqrt{B_1}} + \frac{(\gamma - 3\gamma B_1)\, x_1\, \eta_{n\$13472[2]}\, \xi_{n\$13472[1]}^2}{2\sqrt{B_1}} + \frac{(\gamma - 4\gamma B_1 + 3\gamma B_1^2)\, \eta_{n\$13472[2]}^2\, \xi_{n\$13472[1]}^2}{4\hbar\sqrt{B_1}}\right)\epsilon + O[\epsilon]^2\Big\}$$

$$\mathrm{am} \to \mathbb{E}_{\{i,j\}\to\{k\}}\left[a_k\,(\alpha_i+\alpha_j),\ x_k\left(e^{-\gamma\alpha_j}\,\xi_i+\xi_j\right),\ 1\right]$$

$$\mathrm{bm} \to \mathbb{E}_{\{i,j\}\to\{k\}}\left[b_k\,(\beta_i+\beta_j),\ y_k\,(\eta_i+\eta_j),\ 1-y_k\,\beta_i\,\eta_j\,\epsilon+O[\epsilon]^2\right]$$

$$\mathrm{dm} \to \mathbb{E}_{\{i,j\}\to\{k\}}\Big[a_k\,\alpha_i+a_k\,\alpha_j+b_k\,\beta_i+b_k\,\beta_j,\ y_k\,\eta_i+\frac{y_k\,\eta_j}{\mathcal{A}_i}+\frac{x_k\,\xi_i}{\mathcal{A}_j}+\frac{(1-B_k)\,\eta_j\,\xi_i}{\hbar}+x_k\,\xi_j,$$
$$1+\left(-\frac{y_k\,\beta_i\,\eta_j}{\mathcal{A}_i}-\frac{x_k\,\beta_j\,\xi_i}{\mathcal{A}_j}+a_k\,B_k\,\eta_j\,\xi_i+\frac{\gamma\,\hbar\,x_k\,y_k\,\eta_j\,\xi_i}{\mathcal{A}_i\,\mathcal{A}_j}+\frac{(\gamma-3\,\gamma\,B_k)\,y_k\,\eta_j^2\,\xi_i}{2\,\mathcal{A}_i}+\frac{(\gamma-3\,\gamma\,B_k)\,x_k\,\eta_j\,\xi_i^2}{2\,\mathcal{A}_j}+\frac{\left(\gamma-4\,\gamma\,B_k+3\,\gamma\,B_k^2\right)\eta_j^2\,\xi_i^2}{4\,\hbar}\right)\epsilon+$$
$$O[\epsilon]^2\Big]$$

$$\mathrm{R} \to \mathbb{E}_{\{\}\to\{i,j\}}\left[\hbar\,a_j\,b_i,\ \hbar\,x_j\,y_i,\ 1-\frac{1}{4}\left(\gamma\,\hbar^3\,x_j^2\,y_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{\mathrm{R}} \to \mathbb{E}_{\{\}\to\{i,j\}}\left[-\hbar\,a_j\,b_i,\ -\frac{\hbar\,x_j\,y_i}{B_i},\ 1+\left(-\frac{\hbar^2\,a_j\,x_j\,y_i}{B_i}-\frac{3\,\gamma\,\hbar^3\,x_j^2\,y_i^2}{4\,B_i^2}\right)\epsilon+O[\epsilon]^2\right]$$

$$\mathrm{P} \to \mathbb{E}_{\{i,j\}\to\{\}}\left[\frac{\alpha_j\,\beta_i}{\hbar},\ \frac{\eta_i\,\xi_j}{\hbar},\ 1+\frac{\gamma\,\eta_i^2\,\xi_j^2\,\epsilon}{4\,\hbar}+O[\epsilon]^2\right]$$

$$\mathrm{aS} \to \mathbb{E}_{\{i\}\to\{i\}}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1+\left(-\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-\frac{1}{2}\,\gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{\mathrm{aS}} \to \mathbb{E}_{\{i\}\to\{i\}}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1+\left(\gamma\,\hbar\,x_i\,\mathcal{A}_i\,\xi_i-\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-\frac{1}{2}\,\gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\mathrm{bS} \to \mathbb{E}_{\{i\}\to\{i\}}\left[-b_i\,\beta_i,\ -\frac{y_i\,\eta_i}{B_i},\ 1+\left(-\frac{y_i\,\beta_i\,\eta_i}{B_i}-\frac{\gamma\,\hbar\,y_i^2\,\eta_i^2}{2\,B_i^2}\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{\mathrm{bS}} \to \mathbb{E}_{\{i\}\to\{i\}}\left[-b_i\,\beta_i,\ -\frac{y_i\,\eta_i}{B_i},\ 1+\left(\frac{\gamma\,\hbar\,y_i\,\eta_i}{B_i}-\frac{y_i\,\beta_i\,\eta_i}{B_i}-\frac{\gamma\,\hbar\,y_i^2\,\eta_i^2}{2\,B_i^2}\right)\epsilon+O[\epsilon]^2\right]$$

$$\mathrm{dS} \to \mathbb{E}_{\{i\}\to\{i\}}\Big[-a_i\,\alpha_i-b_i\,\beta_i,\ -\frac{y_i\,\mathcal{A}_i\,\eta_i}{B_i}-x_i\,\mathcal{A}_i\,\xi_i+\frac{(\mathcal{A}_i-B_i\,\mathcal{A}_i)\,\eta_i\,\xi_i}{\hbar\,B_i},$$

Out[∘]=

$$1+\Big(\frac{\gamma\,\hbar\,y_i\,\mathcal{A}_i\,\eta_i}{B_i}-\frac{y_i\,\mathcal{A}_i\,\beta_i\,\eta_i}{B_i}-\frac{\gamma\,\hbar\,y_i^2\,\mathcal{A}_i^2\,\eta_i^2}{2\,B_i^2}-\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-x_i\,\mathcal{A}_i\,\beta_i\,\xi_i+\frac{a_i\,\mathcal{A}_i\,\eta_i\,\xi_i}{B_i}-$$
$$\frac{\gamma\,\hbar\,x_i\,y_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i}{B_i}+\frac{(-\gamma\,\mathcal{A}_i+\gamma\,B_i\,\mathcal{A}_i)\,\eta_i\,\xi_i}{B_i}+\frac{(\mathcal{A}_i-B_i\,\mathcal{A}_i)\,\beta_i\,\eta_i\,\xi_i}{\hbar\,B_i}+\frac{y_i\left(3\,\gamma\,\mathcal{A}_i^2-\gamma\,B_i\,\mathcal{A}_i^2\right)\eta_i^2\,\xi_i}{2\,B_i^2}-$$
$$\frac{1}{2}\,\gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2+\frac{x_i\left(3\,\gamma\,\mathcal{A}_i^2-\gamma\,B_i\,\mathcal{A}_i^2\right)\eta_i\,\xi_i^2}{2\,B_i}+\frac{\left(-3\,\gamma\,\mathcal{A}_i^2+4\,\gamma\,B_i\,\mathcal{A}_i^2-\gamma\,B_i^2\,\mathcal{A}_i^2\right)\eta_i^2\,\xi_i^2}{4\,\hbar\,B_i^2}\Big)\epsilon+O[\epsilon]^2\Big]$$

$$\mathrm{a}\triangle \to \mathbb{E}_{\{i\}\to\{j,k\}}\left[a_j\,\alpha_i+a_k\,\alpha_i,\ x_j\,\xi_i+x_k\,\xi_i,\ 1+\left(-\hbar\,a_j\,x_k\,\xi_i+\frac{1}{2}\,\gamma\,\hbar\,x_j\,x_k\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\mathrm{b}\triangle \to \mathbb{E}_{\{i\}\to\{j,k\}}\left[b_j\,\beta_i+b_k\,\beta_i,\ B_k\,y_j\,\eta_i+y_k\,\eta_i,\ 1+\frac{1}{2}\,\gamma\,\hbar\,B_k\,y_j\,y_k\,\eta_i^2\,\epsilon+O[\epsilon]^2\right]$$

$$\mathrm{d}\triangle \to \mathbb{E}_{\{i\}\to\{j,k\}}\Big[a_j\,\alpha_i+a_k\,\alpha_i+b_j\,\beta_i+b_k\,\beta_i,\ y_j\,\eta_i+B_j\,y_k\,\eta_i+x_j\,\xi_i+x_k\,\xi_i,$$
$$1+\left(\frac{1}{2}\,\gamma\,\hbar\,B_j\,y_j\,y_k\,\eta_i^2-\hbar\,a_j\,x_k\,\xi_i+\frac{1}{2}\,\gamma\,\hbar\,x_j\,x_k\,\xi_i^2\right)\epsilon+O[\epsilon]^2\Big]$$

$$\mathrm{C} \to \mathbb{E}_{\{\}\to\{i\}}\left[0,\ 0,\ \sqrt{B_i}-\frac{1}{2}\left(\hbar\,a_i\,\sqrt{B_i}\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{\mathrm{C}} \to \mathbb{E}_{\{\}\to\{i\}}\left[0,\ 0,\ \frac{1}{\sqrt{B_i}}+\frac{\hbar\,a_i\,\epsilon}{2\,\sqrt{B_i}}+O[\epsilon]^2\right]$$

$$\mathrm{Kink} \to \mathbb{E}_{\{\}\to\{i\}}\left[\hbar\,a_i\,b_i,\ \hbar\,x_i\,y_i,\ \frac{1}{\sqrt{B_i}}+\left(\frac{\hbar\,a_i}{2\,\sqrt{B_i}}-\frac{\gamma\,\hbar^3\,x_i^2\,y_i^2}{4\,\sqrt{B_i}}\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{\mathrm{Kink}} \to \mathbb{E}_{\{\}\to\{i\}}\left[-\hbar\,a_i\,b_i,\ -\frac{\hbar\,x_i\,y_i}{B_i},\ \sqrt{B_i}+\left(-\frac{1}{2}\,\hbar\,a_i\,\sqrt{B_i}-\frac{\hbar^2\,a_i\,x_i\,y_i}{\sqrt{B_i}}-\frac{3\,\gamma\,\hbar^3\,x_i^2\,y_i^2}{4\,B_i^{3/2}}\right)\epsilon+O[\epsilon]^2\right]$$

$$\mathrm{b2t} \to \mathbb{E}_{\{i\}\to\{i\}}\left[a_i\,\alpha_i-\frac{t_i\,\beta_i}{\gamma},\ y_i\,\eta_i+x_i\,\xi_i,\ 1+\frac{a_i\,\beta_i\,\epsilon}{\gamma}+O[\epsilon]^2\right]$$

$$\mathrm{t2b} \to \mathbb{E}_{\{i\}\to\{i\}}\left[a_i\,\alpha_i-\gamma\,b_i\,\tau_i,\ y_i\,\eta_i+x_i\,\xi_i,\ 1+a_i\,\tau_i\,\epsilon+O[\epsilon]^2\right]$$

Check that on the generators this agrees with our conventions in the handout:

```
In[∘]:= Timing@
  {{"[a,x]" → ((𝔼{}→{1,2}[0, 0, a2 x1] // am1,2→1)[3] - (𝔼{}→{1,2}[0, 0, a1 x2] // am1,2→1)[3]),
     "[b,y]" → ((𝔼{}→{1,2}[0, 0, y2 b1] // bm1,2→1)[3] - (𝔼{}→{1,2}[0, 0, y1 b2] // bm1,2→1)[3])} /.
    z_1 → z,
   {"Δ[y]" → Last[𝔼{}→{1}[0, 0, y1] ~B1~ bΔ1→1,2],
    "Δ[b]" → Last[𝔼{}→{1}[0, 0, b1] ~B1~ bΔ1→1,2],
    "Δ[a]" → Last[𝔼{}→{1}[0, 0, a1] ~B1~ aΔ1→1,2],
    "Δ[x]" → Last[𝔼{}→{1}[0, 0, x1] ~B1~ aΔ1→1,2]},
   {
     "S(a)" → ((𝔼{}→{1}[0, 0, a1] ~B1~ aS1)[3]),
     "S(x)" → ((𝔼{}→{1}[0, 0, x1] ~B1~ aS1)[3]),
     "S(b)" → ((𝔼{}→{1}[0, 0, b1] ~B1~ bS1)[3]),
     "S(y)" → ((𝔼{}→{1}[0, 0, y1] ~B1~ bS1)[3])
   } /. z_1 → z}
```

Out[∘]= $\{0.796875, \{\{[a,x] \to -x\gamma,\ [b,y] \to -y\epsilon + O[\epsilon]^3\}, \{\Delta[y] \to (B_2 y_1 + y_2) + O[\epsilon]^3, \Delta[b] \to (b_1 + b_2) + O[\epsilon]^3, \Delta[a] \to (a_1 + a_2) + O[\epsilon]^3, \Delta[x] \to (x_1 + x_2) - \hbar a_1 x_2 \epsilon + \frac{1}{2}\hbar^2 a_1^2 x_2 \epsilon^2 + O[\epsilon]^3\}, \{S(a) \to -a + O[\epsilon]^3, S(x) \to -x - a x \hbar \epsilon - \frac{1}{2}(a^2 x \hbar^2)\epsilon^2 + O[\epsilon]^3, S(b) \to -b + O[\epsilon]^3, S(y) \to -\frac{y}{B} + O[\epsilon]^3\}\}\}$

Hopf algebra axioms on both sides separately.

Associativity of am and bm:

```
In[∘]:= Timing@Block[{$k = 3},
   HL /@ {(am1,2→1 // am1,3→1) ≡ (am2,3→2 // am1,2→1), (bm1,2→1 // bm1,3→1) ≡ (bm2,3→2 // bm1,2→1)}
  ]
```

Out[∘]= {0.171875, {True, True}}

R and *P* are inverses:

```
In[∘]:= Timing@Block[{$k = 3}, {Ri,j, Pi,k, HL[(Ri,j // Pi,k) ≡ 𝔼{k}→{j}[aj αk, xj ξk, 1]]}]
```

Out[∘]= $\{0.15625, \{\mathbb{E}_{\{\}\to\{i,j\}}[\hbar a_j b_i, \hbar x_j y_i, 1 - \frac{1}{4}(\gamma\hbar^3 x_j^2 y_i^2)\epsilon + (\frac{1}{9}\gamma^2\hbar^5 x_j^3 y_i^3 + \frac{1}{32}\gamma^2\hbar^6 x_j^4 y_i^4)\epsilon^2 + (\frac{1}{48}\gamma^3\hbar^5 x_j^2 y_i^2 - \frac{1}{16}\gamma^3\hbar^7 x_j^4 y_i^4 - \frac{1}{36}\gamma^3\hbar^8 x_j^5 y_i^5 - \frac{1}{384}\gamma^3\hbar^9 x_j^6 y_i^6)\epsilon^3 + O[\epsilon]^4],$

$\mathbb{E}_{\{i,k\}\to\{\}}[\frac{\alpha_k\beta_i}{\hbar}, \frac{\eta_i\xi_k}{\hbar}, 1 + \frac{\gamma\eta_i^2\xi_k^2\epsilon}{4\hbar} + \frac{(36\gamma^2\hbar^2\eta_i^2\xi_k^2 + 40\gamma^2\hbar\eta_i^3\xi_k^3 + 9\gamma^2\eta_i^4\xi_k^4)\epsilon^2}{288\hbar^2} + (\frac{1}{24}\gamma^3\hbar\eta_i^2\xi_k^2 + \frac{1}{6}\gamma^3\eta_i^3\xi_k^3 + \frac{13\gamma^3\eta_i^4\xi_k^4}{96\hbar} + \frac{5\gamma^3\eta_i^5\xi_k^5}{144\hbar^2} + \frac{\gamma^3\eta_i^6\xi_k^6}{384\hbar^3})\epsilon^3 + O[\epsilon]^4], \text{True}\}\}$

as and $\overline{\text{aS}}$ are inverses, bs and $\overline{\text{bS}}$ are inverses:

```
In[∘]:= Timing[HL /@ {(aS̄1 // aS1) ≡ 𝔼{1}→{1}[a1 α1, x1 ξ1, 1], (bS̄1 // bS1) ≡ 𝔼{1}→{1}[b1 β1, y1 η1, 1]}]
```

Out[∘]= {0.375, {True, True}}

(co)-associativity on both sides

In[∘]:= Timing[
HL /@ {(a$\Delta_{1\to1,2}$ // a$\Delta_{2\to2,3}$) ≡ (a$\Delta_{1\to1,3}$ // a$\Delta_{1\to1,2}$), (b$\Delta_{1\to1,2}$ // b$\Delta_{2\to2,3}$) ≡ (b$\Delta_{1\to1,3}$ // b$\Delta_{1\to1,2}$),
(am$_{1,2\to1}$ // am$_{1,3\to1}$) ≡ (am$_{2,3\to2}$ // am$_{1,2\to1}$), (bm$_{1,2\to1}$ // bm$_{1,3\to1}$) ≡ (bm$_{2,3\to2}$ // bm$_{1,2\to1}$)}]

Out[∘]= {0.3125, {True, True, True, True}}

Δ is an algebra morphism

In[∘]:= Timing[HL /@ {(am$_{1,2\to1}$ // a$\Delta_{1\to1,2}$) ≡ ((a$\Delta_{1\to1,3}$ a$\Delta_{2\to2,4}$) // (am$_{3,4\to2}$ am$_{1,2\to1}$)),
(bm$_{1,2\to1}$ // b$\Delta_{1\to1,2}$) ≡ ((b$\Delta_{1\to1,3}$ b$\Delta_{2\to2,4}$) // (bm$_{3,4\to2}$ bm$_{1,2\to1}$))}]

Out[∘]= {0.46875, {True, True}}

An explicit formula for aS$_i$

In[∘]:= Timing@Block[{$k = 4}, HL[aS$_i$ ≡ ($\mathbb{E}_{\{i\}\to\{i,j\}}$[$-\alpha_i$ a$_j$, $-\xi_i$ x$_i$,
Sum[Expand[$\frac{e^{\xi_i x_i}(-\hbar\gamma\epsilon)^k}{2^k k!}$ Nest[Expand[$x_i^2\,\partial_{\{x_i,2\}}$ #] &, $e^{-\xi_i e^{\hbar\epsilon a_i} x_i}$, k]], {k, 0, $k}]]$_{\$k}$ //
am$_{i,j\to i}$)]]

Out[∘]= {3.14063, True}

S is convolution inverse of id

In[∘]:= Timing[HL[# ≡ $\mathbb{E}_{\{1\}\to\{1\}}$[0, 0, 1]] & /@ {
(a$\Delta_{1\to1,2}$ ~ B$_1$ ~ aS$_1$) ~ B$_{1,2}$ ~ am$_{1,2\to1}$, (a$\Delta_{1\to1,2}$ ~ B$_2$ ~ aS$_2$) ~ B$_{1,2}$ ~ am$_{1,2\to1}$,
(b$\Delta_{1\to1,2}$ ~ B$_1$ ~ bS$_1$) ~ B$_{1,2}$ ~ bm$_{1,2\to1}$, (b$\Delta_{1\to1,2}$ ~ B$_2$ ~ bS$_2$) ~ B$_{1,2}$ ~ bm$_{1,2\to1}$}]

Out[∘]= {0.453125, {True, True, True, True}}

But not with the opposite product:

In[∘]:= Timing[Short[# ≡ $\mathbb{E}_{\{1\}\to\{1\}}$[0, 0, 1]] & /@ {
(a$\Delta_{1\to1,2}$ ~ B$_1$ ~ aS$_1$) ~ B$_{1,2}$ ~ am$_{2,1\to1}$, (a$\Delta_{1\to1,2}$ ~ B$_2$ ~ aS$_2$) ~ B$_{1,2}$ ~ am$_{2,1\to1}$,
(b$\Delta_{1\to1,2}$ ~ B$_1$ ~ bS$_1$) ~ B$_{1,2}$ ~ bm$_{2,1\to1}$, (b$\Delta_{1\to1,2}$ ~ B$_2$ ~ bS$_2$) ~ B$_{1,2}$ ~ bm$_{2,1\to1}$}]

Out[∘]= $\Big\{0.53125,\ \Big\{\frac{1}{2}\left(-2\gamma\epsilon\hbar x_1 \mathcal{A}_1 \xi_1 + \gamma^2\epsilon^2\hbar^2 x_1 \mathcal{A}_1 \xi_1 - 2\gamma\epsilon^2\hbar^2 a_1 x_1 \mathcal{A}_1 \xi_1 + 2\gamma^2\epsilon^2\hbar^2 x_1^2 \mathcal{A}_1^2 \xi_1^2\right) == 0,$

$\frac{1}{2}\left(-2\gamma\epsilon\hbar x_1 \xi_1 - \gamma^2\epsilon^2\hbar^2 x_1 \xi_1 + 2\gamma^2\epsilon^2\hbar^2 x_1^2 \xi_1^2\right) == 0,$

$\frac{1}{2}\left(-2\gamma\epsilon\hbar y_1 \eta_1 - \gamma^2\epsilon^2\hbar^2 y_1 \eta_1 + 2\gamma^2\epsilon^2\hbar^2 y_1^2 \eta_1^2\right) == 0,$

$\frac{-2\gamma\epsilon\hbar B_1 y_1 \eta_1 + \ll 3 \gg + 2\gamma^2\epsilon^2\hbar^2 y_1^2 \eta_1^2}{2 B_1^2} == 0\Big\}\Big\}$

S is an algebra anti-(co)morphism

In[∘]:= Timing[HL /@ {am$_{1,2\to1}$ ~ B$_1$ ~ aS$_1$ ≡ (aS$_1$ aS$_2$) ~ B$_{1,2}$ ~ am$_{2,1\to1}$, bm$_{1,2\to1}$ ~ B$_1$ ~ bS$_1$ ≡ (bS$_1$ bS$_2$) ~ B$_{1,2}$ ~ bm$_{2,1\to1}$,
aS$_1$ ~ B$_1$ ~ a$\Delta_{1\to1,2}$ ≡ a$\Delta_{1\to2,1}$ ~ B$_{1,2}$ ~ (aS$_1$ aS$_2$), bS$_1$ ~ B$_1$ ~ b$\Delta_{1\to1,2}$ ≡ b$\Delta_{1\to2,1}$ ~ B$_{1,2}$ ~ (bS$_1$ bS$_2$)}]

Out[∘]= {0.640625, {True, True, True, True}}

Pairing axioms

```
In[]:= Timing[HL /@ {(bm1,2→1 𝔼{3}→{3}[α3 a3, ξ3 x3, 1]) ~B1,3~P1,3 ≡
         (𝔼{1}→{1}[β1 b1, η1 y1, 1] 𝔼{2}→{2}[β2 b2, η2 y2, 1] aΔ3→4,5) ~B1,4~P1,4~B2,5~P2,5,
       (bΔ1→1,2 𝔼{3}→{3}[α3 a3, ξ3 x3, 1] 𝔼{4}→{4}[α4 a4, ξ4 x4, 1]) ~B1,3~P1,3~B2,4~P2,4 ≡
         (𝔼{1}→{1}[β1 b1, η1 y1, 1] am3,4→3) ~B1,3~P1,3 }]
Out[]= {0.3125, {True, True}}
```

```
In[]:= Timing[HL /@ {((bS1 𝔼{2}→{2}[α2 a2, ξ2 x2, 1]) // P1,2) ≡ ((𝔼{1}→{1}[β1 b1, η1 y1, 1] aS2) // P1,2),
       (b̄S̄1 𝔼{2}→{2}[α2 a2, ξ2 x2, 1]) ~B1,2~P1,2 ≡ (𝔼{1}→{1}[β1 b1, η1 y1, 1] āS̄2) ~B1,2~P1,2}]
Out[]= {0.21875, {True, True}}
```

Tests for the double.

Check the double formulas on the generators agree with SL2Portfolio.pdf:

```
In[]:= Timing@{{
          "[a,y]" →
           ((𝔼{}→{1,2}[0, 0, y2 a1] ~B1,2~dm1,2→1) [3] - (𝔼{}→{1,2}[0, 0, y1 a2] ~B1,2~dm1,2→1) [3]),
          "[b,x]" → ((𝔼{}→{1,2}[0, 0, x2 b1] ~B1,2~dm1,2→1) [3] -
             (𝔼{}→{1,2}[0, 0, x1 b2] ~B1,2~dm1,2→1) [3]),
          "xy-qyx" → ((𝔼{}→{1,2}[0, 0, x1 y2] ~B1,2~dm1,2→1) [3] -
             (1 + ϵ) (𝔼{}→{1,2}[0, 0, y1 x2] ~B1,2~dm1,2→1) [3])
         } /. {z_1 → z} // Expand // Factor,
      {
        "Δ(a)" → ((𝔼{}→{1}[0, 0, a1] ~B1~dΔ1→1,2) [3]),
        "Δ(x)" → ((𝔼{}→{1}[0, 0, x1] ~B1~dΔ1→1,2) [3]),
        "Δ(b)" → ((𝔼{}→{1}[0, 0, b1] ~B1~dΔ1→1,2) [3]),
        "Δ(y)" → ((𝔼{}→{1}[0, 0, y1] ~B1~dΔ1→1,2) [3])
       } // Simplify,
      {
         "S(a)" → ((𝔼{}→{1}[0, 0, a1] ~B1~dS1) [3]),
         "S(x)" → ((𝔼{}→{1}[0, 0, x1] ~B1~dS1) [3]),
         "S(b)" → ((𝔼{}→{1}[0, 0, b1] ~B1~dS1) [3]),
         "S(y)" → ((𝔼{}→{1}[0, 0, y1] ~B1~dS1) [3])
        } /. {z_1 → z} // Simplify
     }
```

Out[]= $\{3.51563, \{\{[a,y] \to -y\,\gamma + O[\epsilon]^3,\ [b,x] \to x\,\epsilon + O[\epsilon]^3,$

$\text{xy-qyx} \to \frac{1-B}{\hbar} + (a\,B - x\,y + x\,y\,\gamma\,\hbar)\,\epsilon + \left(-\frac{1}{2}a^2\,B\,\hbar + \frac{1}{2}x\,y\,\gamma^2\,\hbar^2\right)\epsilon^2 + O[\epsilon]^3\},$

$\{\Delta(a) \to (a_1 + a_2) + O[\epsilon]^3,\ \Delta(x) \to (x_1 + x_2) - \hbar\,a_1\,x_2\,\epsilon + \frac{1}{2}\hbar^2\,a_1^2\,x_2\,\epsilon^2 + O[\epsilon]^3,$

$\Delta(b) \to (b_1 + b_2) + O[\epsilon]^3,\ \Delta(y) \to (y_1 + B_1\,y_2) + O[\epsilon]^3\},$

$\{S(a) \to -a + O[\epsilon]^3,\ S(x) \to -x - a\,x\,\hbar\,\epsilon - \frac{1}{2}\left(a^2\,x\,\hbar^2\right)\epsilon^2 + O[\epsilon]^3,$

$S(b) \to -b + O[\epsilon]^3,\ S(y) \to -\frac{y}{B} + \frac{y\,\gamma\,\hbar\,\epsilon}{B} - \frac{\left(y\,\gamma^2\,\hbar^2\right)\epsilon^2}{2\,B} + O[\epsilon]^3\}\}\}$

(co)-associativity

In[∘]:= $\mathtt{Timing}\big[\mathtt{HL}\ /@\ \{(d\Delta_{1\to1,2}\ //\ d\Delta_{2\to2,3}) \equiv (d\Delta_{1\to1,3}\ //\ d\Delta_{1\to1,2}),\ (dm_{1,2\to1}\ //\ dm_{1,3\to1}) \equiv (dm_{2,3\to2}\ //\ dm_{1,2\text{->}1})\}\big]$

Out[∘]= {1.85938, {True, True}}

Δ is an algebra morphism

In[∘]:= $\mathtt{Timing}@\mathtt{HL}\big[dm_{1,2\to1} \sim B_1 \sim d\Delta_{1\to1,2} \equiv (d\Delta_{1\to1,3}\ d\Delta_{2\to2,4}) \sim B_{1,2,3,4} \sim (dm_{3,4\to2}\ dm_{1,2\to1})\big]$

Out[∘]= {1.875, True}

S_2 inverts R, but not S_1:

In[∘]:= $\mathtt{Timing}@\{R_{1,2} \sim B_1 \sim dS_1 \equiv \overline{R}_{1,2},\ \mathtt{HL}[R_{1,2} \sim B_2 \sim dS_2 \equiv \overline{R}_{1,2}]\}$

Out[∘]= $\Big\{0.359375,\ \Big\{\frac{1}{4 B_1^3}\big(4\gamma\epsilon\hbar^2 B_1^2 x_2 y_1 - 2\gamma^2\epsilon^2\hbar^3 B_1^2 x_2 y_1 + 4\gamma\epsilon^2\hbar^3 a_2 B_1^2 x_2 y_1 + 8\gamma^2\epsilon^2\hbar^4 B_1 x_2^2 y_1^2 - 4\gamma\epsilon^2\hbar^4 a_2 B_1 x_2^2 y_1^2 - 3\gamma^2\epsilon^2\hbar^5 x_2^3 y_1^3\big) == 0,\ \mathtt{True}\Big\}\Big\}$

S is convolution inverse of id

In[∘]:= $\mathtt{Timing}\big[\mathtt{HL}[\#\equiv \mathbb{E}_{\{1\}\to\{1\}}[0,0,1]]\ \&\ /@\ \{(d\Delta_{1\to1,2} \sim B_1 \sim dS_1) \sim B_{1,2} \sim dm_{1,2\to1},\ (d\Delta_{1\to1,2} \sim B_2 \sim dS_2)\ //\ dm_{1,2\to1}\}\big]$

Out[∘]= {3.51563, {True, True}}

S is a (co)-algebra anti-morphism

In[∘]:= $\mathtt{Timing}\big[\mathtt{HL}\ /@\ \mathtt{Expand}\ /@\ \{dm_{1,2\to1} \sim B_1 \sim dS_1 \equiv (dS_1\ dS_2) \sim B_{1,2} \sim dm_{2,1\to1},\ dS_1 \sim B_1 \sim d\Delta_{1\to1,2} \equiv d\Delta_{1\to2,1} \sim B_{1,2} \sim (dS_1\ dS_2)\}\big]$

Out[∘]= {7.4375, {True, True}}

Quasi-triangular axiom 1:

In[∘]:= $\mathtt{Timing}@\mathtt{HL}[R_{1,2} \sim B_1 \sim d\Delta_{1\text{->}1,3} \equiv (R_{1,4}\ R_{3,2}) \sim B_{2,4} \sim dm_{2,4\to2}]$

Out[∘]= {0.171875, True}

Quasi-triangular axiom 2:

In[∘]:= $\mathtt{Timing}@\mathtt{HL}\big[((d\Delta_{1\text{->}1,2}\ R_{3,4}) \sim B_{1,2,3,4} \sim (dm_{1,3\to1}\ dm_{2,4\to2})) \equiv ((d\Delta_{1\to2,1}\ R_{3,4}) \sim B_{1,2,3,4} \sim (dm_{3,1\to1}\ dm_{4,2\to2}))\big]$

Out[∘]= {1.59375, True}

The Drinfel'd element inverse property, $(u_1\,\overline{u}_2) \sim B_{1,2} \sim \mathrm{dm}_{1,2\to1} \equiv \mathbb{E}[0,0,1]$:

In[∘]:= $\mathtt{Timing}@\mathtt{HL}\big[((R_{1,2} \sim B_1 \sim dS_1 \sim B_{1,2} \sim dm_{2,1\to i})\ (R_{1,2} \sim B_2 \sim dS_2 \sim B_2 \sim dS_2 \sim B_{1,2} \sim dm_{2,1\to j})) \sim B_{i,j} \sim dm_{i,j\to i} \equiv \mathbb{E}_{\{\}\to\{i\}}[0,0,1]\big]$

Out[∘]= {1.51563, True}

The ribbon element v satisfies $v^2 = S(u)\,u$. The spinner C=uv^{-1}. It is convenient to compute $z = S(u)\,u^{-1}$ which is something easy.

In[∘]:= $\texttt{Timing@Block}\big[\{\$k = 2\},\ \big(\big(\big(R_{1,2} \sim B_1 \sim dS_1 \sim B_{1,2} \sim dm_{2,1\to i}\big) \sim B_i \sim dS_i\big)\ \big(R_{1,2} \sim B_2 \sim dS_2 \sim B_2 \sim dS_2 \sim B_{1,2} \sim dm_{2,1\to j}\big)\big) \sim B_{i,j} \sim dm_{i,j\to i}\big]$

Out[∘]= $\left\{2.10938,\ \mathbb{E}_{\{\}\to\{i\}}\left[0, 0, \frac{1}{B_i} + \frac{\hbar\, a_i\, \epsilon}{B_i} + \frac{\hbar^2\, a_i^2\, \epsilon^2}{2\, B_i} + O[\epsilon]^3\right]\right\}$

In[∘]:= $\texttt{Timing@Block}\big[\{\$k = 2\},\ \texttt{HL /@}\big\{\big(C_i\, \overline{C}_j\big) \sim B_{i,j} \sim dm_{i,j\to i} \equiv \mathbb{E}_{\{\}\to\{i\}}[0, 0, 1],\ \big(\overline{C}_i\, C_j\big) \sim B_{i,j} \sim dm_{i,j\to i} \equiv \big(\big(\big(R_{1,2} \sim B_1 \sim dS_1 \sim B_{1,2} \sim dm_{2,1\to i}\big) \sim B_i \sim dS_i\big)\ \big(R_{1,2} \sim B_2 \sim dS_2 \sim B_2 \sim dS_2 \sim B_{1,2} \sim dm_{2,1\to j}\big)\big) \sim B_{i,j} \sim dm_{i,j\to i}\big\}\big]$

Out[∘]= {2.40625, {True, True}}

Reidemeister 2:

In[∘]:= $\texttt{Timing}\big[\texttt{HL}[\# \equiv \mathbb{E}_{\{\}\to\{1,2\}}[0, 0, 1]]\ \&\ \texttt{/@}\ \big\{\big(\overline{R}_{1,2}\, R_{3,4}\big) \sim B_{1,2,3,4} \sim \big(dm_{1,3\to 1}\, dm_{2,4\to 2}\big),\ \big(R_{1,2}\, \overline{R}_{3,4}\big) \sim B_{1,2,3,4} \sim \big(dm_{1,3\to 1}\, dm_{2,4\to 2}\big)\big\}\big]$

Out[∘]= {1.29688, {True, True}}

Cyclic Reidemeister 2:

In[∘]:= $\texttt{Timing@HL}\big[\big(R_{1,4}\, \overline{R}_{5,2}\, \overline{C}_3\big) \sim B_{2,4} \sim dm_{2,4\to 2} \sim B_{1,3} \sim dm_{1,3\to 1} \sim B_{1,5} \sim dm_{1,5\to 1} \equiv \overline{C}_1\, \mathbb{E}_{\{\}\to\{2\}}[0, 0, 1]\big]$

Out[∘]= {0.859375, True}

Reidemeister 3:

In[∘]:= $\texttt{Timing@HL}\big[\big((R_{1,2}\, R_{4,3}\, R_{5,6}) \sim B_{1,4} \sim dm_{1,4\to 1} \sim B_{2,5} \sim dm_{2,5\to 2} \sim B_{3,6} \sim dm_{3,6\to 3}\big) \equiv \big((R_{1,6}\, R_{2,3}\, R_{4,5}) \sim B_{1,4} \sim dm_{1,4\to 1} \sim B_{2,5} \sim dm_{2,5\to 2} \sim B_{3,6} \sim dm_{3,6\to 3}\big)\big]$

Out[∘]= {1.25, True}

Relations between the four kinks:

In[∘]:= $\texttt{Timing}\big[\texttt{HL /@}\big\{Kink_i \equiv \big(R_{3,1}\, C_2\big) \sim B_{1,2} \sim dm_{1,2\to 1} \sim B_{1,3} \sim dm_{1,3\to i},\ \overline{Kink}_j \equiv \big(\overline{R}_{3,1}\, \overline{C}_2\big) \sim B_{1,2} \sim dm_{1,2\to 1} \sim B_{1,3} \sim dm_{1,3\to j},\ \big(Kink_i\, \overline{Kink}_j\big) \sim B_{i,j} \sim dm_{i,j\to 1} \equiv \mathbb{E}_{\{\}\to\{1\}}[0, 0, 1]\big\}\big]$

Out[∘]= {2.42188, {True, True, True}}

The Trefoil

In[∘]:= $\texttt{Timing@Block}\big[\{\$k = 1\},\ Z = R_{1,5}\, R_{6,2}\, R_{3,7}\, \overline{C}_4\, \overline{Kink}_8\, \overline{Kink}_9\, \overline{Kink}_{10};\ \texttt{Do}[Z = Z \sim B_{1,r} \sim dm_{1,r\to 1}, \{r, 2, 10\}];\ \{\texttt{Simplify /@}\, Z,\ \texttt{Simplify /@}\, (Z \sim B_1 \sim b2t_1\ \texttt{/.}\ T_1 \to T)\}\big]$

Out[∘]= $\Big\{2.03125, \Big\{\mathbb{E}_{\{\}\to\{1\}}\Big[0, 0, \frac{B_1}{1 - B_1 + B_1^2} - \frac{\hbar\, B_1\, \big(-a_1\, \big(-1 + B_1 - B_1^3 + B_1^4\big) + \gamma\, \big(B_1 - 2\, B_1^2 - 2\, B_1^4 + 2\, \hbar\, x_1\, y_1 + B_1^3\, \big(3 + 2\, \hbar\, x_1\, y_1\big)\big)\big)\, \epsilon}{\big(1 - B_1 + B_1^2\big)^3} + O[\epsilon]^2\Big],\ \mathbb{E}_{\{\}\to\{1\}}\Big[0, 0, \frac{T}{1 - T + T^2} + \frac{T\, \hbar\, \big(T\, \big(-1 + 2\, T - 3\, T^2 + 2\, T^3\big)\, \gamma + 2\, \big(-1 + T - T^3 + T^4\big)\, a_1 - 2\, \big(1 + T^3\big)\, \gamma\, \hbar\, x_1\, y_1\big)\, \epsilon}{\big(1 - T + T^2\big)^3} + O[\epsilon]^2\Big]\Big\}\Big\}$

Program

In[∘]:=
```
Define[kR_{i,j} = (R_{i,j} // (b2t_i b2t_j)) /. t_{i|j} → t,
  kR̄_{i,j} = (R̄_{i,j} // (b2t_i b2t_j)) /. {t_{i|j} → t, T_{i|j} → T},
  km_{i,j→k} = ((t2b_i t2b_j) // dm_{i,j→k} // b2t_k) /. {t_k → t, T_k → T, τ_{i|j} → 0},
  kC_i = (C_i // b2t_i) /. T_i → T,
  kC̄_i = (C̄_i // b2t_i) /. T_i → T,
  kKink_i = (Kink_i // b2t_i) /. {t_i → t, T_i → T},
  kKink̄_i = (Kink̄_i // b2t_i) /. {t_i → t, T_i → T}]
```

In[∘]:=
```
Timing@Block[{$k = 1},
  Z = kR_{1,5} kR_{6,2} kR_{3,7} kC̄_4 kKink̄_8 kKink̄_9 kKink̄_10;
  Do[Z = Z~B_{1,r}~km_{1,r→1}, {r, 2, 10}];
  Simplify /@ Z]
```

Out[∘]= $\left\{1.23438, \mathbb{E}_{\{\}\to\{1\}}\left[0, 0, \frac{T}{1-T+T^2} + \frac{T\hbar\left(T\left(-1+2T-3T^2+2T^3\right)\gamma+2\left(-1+T-T^3+T^4\right)a_1-2\left(1+T^3\right)\gamma\hbar x_1 y_1\right)\epsilon}{\left(1-T+T^2\right)^3} + O[\epsilon]^2\right]\right\}$

RVK, rot, Z from 2016-09/OneSmidgen.nb. See also local version in this folder.

Program

Some details of the code below are at http://drorbn.net/bbs/show?shot=Dror-160920-151350.jpg.

Program

In[∘]:=
```
RVK::usage =
  "RVK[xs, rots] represents a Rotational Virtual Knot with a list of n Xp/Xm crossings
    xs and a length 2n list of rotation numbers rots. Crossing
    sites are indexed 1 through 2n, and rots⟦k⟧ is the rotation
    between site k-1 and site k. RVK is also a casting operator
    converting to the RVK presentation from other knot presentations.";
```

Program

In[∘]:=
```
RVK[pd_PD] := PP_RVK@Module[{n, xs, x, rots, front = {0}, k},
    n = Length@pd; rots = Table[0, {2 n}];
    xs = Cases[pd, x_X :> { Xp[x⟦4⟧, x⟦1⟧]   PositiveQ@x
                            Xm[x⟦2⟧, x⟦1⟧]   True        ];
    For[k = 0, k < 2 n, ++k, If[k == 0 ∨ FreeQ[front, -k],
      front = Flatten[front /. k → (xs /. {
              Xp[k + 1, l_] | Xm[l_, k + 1] :> {l, k + 1, 1 - l},
              Xp[l_, k + 1] | Xm[k + 1, l_] :> (++rots⟦l⟧; {1 - l, k + 1, l})
            })],
      Cases[front, k | -k] /. {k, -k} :> --rots⟦k + 1⟧;
     ]];
    RVK[xs, rots] ];
RVK[K_] := RVK[PD[K]];
```

In[∘]:=
```
xs = Cases[pd, x_X :> If[PositiveQ@x, Xp[x⟦4⟧, x⟦1⟧], Xm[x⟦2⟧, x⟦1⟧]]];
```

In[◦]:=
```
RVK[Knot[10, 100]]
```

KnotTheory: Loading precomputed data in PD4Knots`.

Out[◦]=
```
RVK[{Xp[1, 6], Xp[5, 18], Xm[13, 20], Xm[7, 14], Xm[3, 10], Xm[9, 16], Xm[11, 4], Xm[15, 8],
  Xm[19, 12], Xp[17, 2]}, {0, 0, 0, 0, 0, -1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0}]
```

Program

In[◦]:=
```
rot[i_, 0] := 𝔼{}→{i}[0, 0, 1];
rot[i_, n_] := Module[{j},
   rot[i, n] = If[n > 0, rot[i, n - 1] kCj, rot[i, n + 1] kC̅j] // kmi,j→i];
```

Program

In[◦]:=
```
Z[K_] := Z[RVK@K];
Z[rvk_RVK] := (*Z[rvk] =*)
 Monitor[ PP"Z"@Module[{todo, n, rots, ζ, done, st, cx, ζ1, i, j, k, k1, k2, k3},
    {todo, rots} = List @@ rvk;
    AppendTo[rots, 0];
    n = Length[todo];
    ζ = 𝔼{}→{0}[0, 0, 1];
    done = {0};
    st = Range[0, 2 n + 1];
    While[{} =!= ($M = todo),
     cx =
      RandomChoice@MaximalBy[todo, Length[done ⋂ {#[[1]], #[[2]], #[[1]] - 1, #[[2]] - 1}] &];
     {i, j} = List @@ cx;
     ζ1 = Switch[Head[cx],
        Xp, (kRi,j kKink̅k) // kmj,k→j,
        Xm, (kR̅i,j kKinkk) // kmj,k→j
       ];
     ζ1 = (rot[k, rots[[i]]] ζ1) // kmk,i→i; rots[[i]] = 0;
     ζ1 = (ζ1 rot[k, rots[[i + 1]]]) // kmi,k→i; rots[[i + 1]] = 0;
     ζ1 = (rot[k, rots[[j]]] ζ1) // kmk,j→j; rots[[j]] = 0;
     ζ1 = (ζ1 rot[k, rots[[j + 1]]]) // kmj,k→j; rots[[j + 1]] = 0;
     ζ *= ζ1;
     If[MemberQ[done, i], ζ = ζ // kmi,i+1→i; st = st /. st[[i + 2]] → st[[i + 1]]];
     If[MemberQ[done, i - 1], ζ = ζ // kmst[[i]],i→st[[i]]; st = st /. st[[i + 1]] → st[[i]]];
     If[MemberQ[done, j], ζ = ζ // kmj,j+1→j; st = st /. st[[j + 2]] → st[[j + 1]]];
     If[MemberQ[done, j - 1], ζ = ζ // kmst[[j]],j→st[[j]]; st = st /. st[[j + 1]] → st[[j]]];
     done = done ⋃ {i - 1, i, j - 1, j};
     todo = DeleteCases[todo, cx]
    ];
    CF /@ (ζ /. {x0 → x, y0 → y, a0 → a})
   ], $M]
```

Knot

In[∘]:= **$k = 1; Timing@Z@Knot[10, 100]**

Knot

Out[∘]= $\{27.3281, \mathbb{E}_{\{\}\to\{0\}}[0, 0, \frac{T^4}{1-4T+9T^2-12T^3+13T^4-12T^5+9T^6-4T^7+T^8} + ((a(-8T^4\hbar+24T^5\hbar-36T^6\hbar+24T^7\hbar-24T^9\hbar+36T^{10}\hbar-24T^{11}\hbar+8T^{12}\hbar))/(1-8T+34T^2-96T^3+203T^4-344T^5+492T^6-608T^7+653T^8-608T^9+492T^{10}-344T^{11}+203T^{12}-96T^{13}+34T^{14}-8T^{15}+T^{16}) + (-6T^4\gamma\hbar+44T^5\gamma\hbar-167T^6\gamma\hbar+410T^7\gamma\hbar-733T^8\gamma\hbar+1016T^9\gamma\hbar-1140T^{10}\gamma\hbar+1048T^{11}\gamma\hbar-776T^{12}\gamma\hbar+440T^{13}\gamma\hbar-156T^{14}\gamma\hbar-16T^{15}\gamma\hbar+79T^{16}\gamma\hbar-70T^{17}\gamma\hbar+37T^{18}\gamma\hbar-12T^{19}\gamma\hbar+2T^{20}\gamma\hbar)/(1-12T+75T^2-316T^3+1002T^4-2544T^5+5394T^6-9840T^7+15771T^8-22512T^9+28866T^{10}-33432T^{11}+35095T^{12}-33432T^{13}+28866T^{14}-22512T^{15}+15771T^{16}-9840T^{17}+5394T^{18}-2544T^{19}+1002T^{20}-316T^{21}+75T^{22}-12T^{23}+T^{24}) + (x\,y(-8T^4\gamma\hbar^2+16T^5\gamma\hbar^2-20T^6\gamma\hbar^2+4T^7\gamma\hbar^2+4T^8\gamma\hbar^2-20T^9\gamma\hbar^2+16T^{10}\gamma\hbar^2-8T^{11}\gamma\hbar^2))/(1-8T+34T^2-96T^3+203T^4-344T^5+492T^6-608T^7+653T^8-608T^9+492T^{10}-344T^{11}+203T^{12}-96T^{13}+34T^{14}-8T^{15}+T^{16}))\,\epsilon+O[\epsilon]^2]\}$

In[∘]:= **$k = 1; Timing@Simplify[Z@Knot[10, 100]]**

Out[∘]= $\{40.9844, \mathbb{E}_{\{\}\to\{0\}}[0, 0, \frac{T^4}{1-4T+9T^2-12T^3+13T^4-12T^5+9T^6-4T^7+T^8} + (T^4\hbar(4a(-2+14T-51T^2+120T^3-203T^4+258T^5-246T^6+152T^7-152T^9+246T^{10}-258T^{11}+203T^{12}-120T^{13}+51T^{14}-14T^{15}+2T^{16}) + \gamma(-6+2T^{16}-8xy\hbar-440T^9(-1+xy\hbar)-4T^{15}(3+2xy\hbar)+8T^8(-97+21xy\hbar)+8T^7(131+21xy\hbar)-20T^6(57+22xy\hbar)+T^{14}(37+48xy\hbar)+T(44+48xy\hbar)-8T^{11}(2+61xy\hbar)+8T^5(127+68xy\hbar)-2T^{13}(35+78xy\hbar)+4T^{10}(-39+136xy\hbar)-T^2(167+156xy\hbar)+T^{12}(79+324xy\hbar)+T^3(410+324xy\hbar)-T^4(733+488xy\hbar)))\,\epsilon)/(1-4T+9T^2-12T^3+13T^4-12T^5+9T^6-4T^7+T^8)^3+O[\epsilon]^2]\}$

In[∘]:= **EndProfile[];**

Profile

In[∘]:= **PrintProfile[]**

Profile

```
Out[∘]= ProfileRoot is root. Profiled time: 84.826
   (      1)    0.110/ 40.360 above Z
   (    157)    0.471/ 35.263 above B
   (     37)    0.158/  9.125 above Boot
   (    147)    0.062/  0.078 above CF
   (      2)        0/      0 above RVK
CF: called 13451 times, time in 26.912/65.31
   (   1047)    0.986/  3.894 under EEQ
   (     47)    0.063/  0.094 under Boot
   (   1347)    6.901/ 19.966 under LZip
   (    147)    0.062/  0.078 under ProfileRoot
   (   9875)   18.680/ 40.837 under QZip
   (    988)    0.220/  0.441 under QZip4
   (  36730)   12.274/ 38.398 above CCF
Together: called 37864 times, time in 19.265/26.374
   (  37864)   19.265/ 26.374 under CCF
   (  37864)    6.155/  7.109 above Exp
```

```
CCF: called 37864 times, time in 12.978/39.352
  ( 36730)  12.274/ 38.398 under CF
  (  1134)   0.704/  0.954 under Exp
  ( 37864)  19.265/ 26.374 above Together
Zip: called 2851 times, time in 8.439/39.908
  (   294)   1.041/  6.525 under LZip
  (   294)   0.812/  4.182 under QZip
  (    44)   0.015/  0.045 under QZip4
  (  2219)   6.571/ 29.156 under Zip
  (  2851)   2.313/  2.313 above Collect
  (  2219)   6.571/ 29.156 above Zip
Exp: called 37864 times, time in 6.155/7.109
  ( 37864)   6.155/  7.109 under Together
  (  1134)   0.704/  0.954 above CCF
LZip: called 294 times, time in 5.275/36.049
  (   294)   5.275/ 36.049 under B
  (  1047)   0.389/  4.283 above EEQ
  (  1347)   6.901/ 19.966 above CF
  (   294)   1.041/  6.525 above Zip
Collect: called 2851 times, time in 2.313/2.313
  (  2851)   2.313/  2.313 under Zip
QZip: called 294 times, time in 1.847/47.43
  (   294)   1.847/ 47.430 under B
  (  9875)  18.680/ 40.837 above CF
  (    22)   0.078/  0.564 above QZip4
  (   294)   0.812/  4.182 above Zip
B: called 294 times, time in 0.674/84.153
  (    72)   0.109/ 40.171 under Z
  (    65)   0.094/  8.719 under Boot
  (   157)   0.471/ 35.263 under ProfileRoot
  (   294)   5.275/ 36.049 above LZip
  (   294)   1.847/ 47.430 above QZip
Boot: called 59 times, time in 0.391/14.048
  (     3)      0/  0.079 under Z
  (    19)   0.233/  4.844 under Boot
  (    37)   0.158/  9.125 under ProfileRoot
  (    65)   0.094/  8.719 above B
  (    19)   0.233/  4.844 above Boot
  (    47)   0.063/  0.094 above CF
EEQ: called 1047 times, time in 0.389/4.283
  (  1047)   0.389/  4.283 under LZip
  (  1047)   0.986/  3.894 above CF
Z: called 1 times, time in 0.11/40.36
  (     1)   0.110/ 40.360 under ProfileRoot
  (    72)   0.109/ 40.171 above B
  (     3)      0/  0.079 above Boot
QZip4: called 22 times, time in 0.078/0.564
  (    22)   0.078/  0.564 under QZip
  (   988)   0.220/  0.441 above CF
  (    44)   0.015/  0.045 above Zip
RVK: called 2 times, time in 0./0.
  (     2)      0/     0 under ProfileRoot
```